Добрый день!
Пытаюсь реализовать программный вывод произвольного количества картинок в определённое место Табличного документа.
Нужно, чтобы фон был прозрачный, то есть, через надписи проступали бы другие элементы табличного документа.
Сами картинки обработал в онлайн-редакторе, чтобы был прозрачный фон. Если вставлять картинки через Конфигуратор в макет ТД, до всё нормально, фон прозрачный. При генерации картинок, всегда выставляю параметр прозрачности в Истина.

```1c
    &НаКлиенте
    Процедура ВставитьФаксимиле(Команда)
    Если не ЗначениеЗаполнено(АдресФайлаФаксимиле) Тогда
    Сообщить("Выберите пример картинки факсимиле!");
    Возврат;
    КонецЕсли;
    СодержимоеКартинки = Новый Картинка(АдресФайлаФаксимиле,Истина);
    ВставитьФаксимилеВТД(СодержимоеКартинки.ПолучитьДвоичныеДанные());
    КонецПроцедуры

    &НаСервере
    Процедура ВставитьФаксимилеВТД(ДД_Факсимиле,Лево = Неопределено, Верх = Неопределено)
    СчётчикКартинок = СчётчикКартинок + 1;
    НоваяФаксимиле = ТабличныйДокумент.Рисунки.Добавить(ТипРисункаТабличногоДокумента.Картинка);
    НоваяФаксимиле.Картинка = Новый Картинка(ДД_Факсимиле,Истина);
    НоваяФаксимиле.Линия = Новый Линия(ТипЛинииРисункаТабличногоДокумента.НетЛинии);
    //НоваяФаксимиле.РазмерКартинки = РазмерКартинки.АвтоРазмер;
    НоваяФаксимиле.Высота = 10;
    НоваяФаксимиле.Ширина = 10;
    НоваяФаксимиле.Имя = "ФаксимилеРазные" + Формат(СчётчикКартинок, "ЧГ=");
    СтрокаТаблицыФаксимиле = ПечатиИПодписиРазные.Добавить();
    СтрокаТаблицыФаксимиле.ПечатьИлиПодпись = ТипФаксимилеРазные;
    СтрокаТаблицыФаксимиле.ИмяКартинкиВТД = НоваяФаксимиле.Имя;
    СтрокаТаблицыФаксимиле.СамаКартинка = НоваяФаксимиле.Картинка;
    СтрокаТаблицыФаксимиле.Координаты = "{0,0}";
    КонецПроцедуры
```

У меня не отрабатывается прозрачность! Вообще никак! Я пытался создавать картинку из файла на клиенте, чтобы передать её на сервер и в ТД непосредственно, в приведённом примере, конвертирую картинку в двоичные данные и обратно. Также, поскольку в синтаксис-помощнике в статье конструктора объекта типа "картинка" есть фраза
<ПрозрачныйФон> (необязательный)
Тип: Булево.
Определяет, будет ли использован цвет прозрачности при создании картинки.
Используется только для картинок, формат которой не поддерживает прозрачности.
То я пробовал не использовать второй параметр в конструкторе картинки.
Не прозрачная, и всё тут.
Кто знает, куда девается прозрачность картинок табличного документа, при каком преобразовании это происходит?
С уважением, я.
---
У меня вот так сделано, и вроде все работает

```1c
    Рисунок=ТабличныйДокумент.Рисунки.Добавить(ТипРисункаТабличногоДокумента.Картинка);
    Рисунок.ЦветФона=Новый Цвет();
    Рисунок.Картинка = Новый Картинка(РезНастройка.Ссылка.Картинка.получить()); // картинка в хранилище в справочнике
    Рисунок.Лево = РезНастройка.Лево;//3
    Рисунок.Верх = РезНастройка.Верх;//ТабличныйДокумент.ВысотаСтраницы-80;
    Рисунок.Высота = РезНастройка.Высота;//50;
    Рисунок.Ширина = РезНастройка.Ширина;//90;
    Рисунок.РазмерКартинки =РазмерКартинки.Растянуть;
    Рисунок.Линия= Новый Линия(ТипЛинииРисункаТабличногоДокумента.НетЛинии);
```

Может надо именно Рисунок.ЦветФона=Новый Цвет(); указать
---
Прозрачность доступны только для картинок форматов BMP, JPG и TIFF
---
о, уау! Я попробую сейчас конвертнуть; похоже, это решение. Спасибо в любом случае.
---
Прозрачность, если мне память не изменяет, можно сделать только в формате png.
---
нет, у меня получилось, с помощью онлайн-конвертеров, выставить также в GIF прозрачность.
Я подчёркиваю, что если добавляю ту же самую картинку в конфигураторе на ТД, то всё ОК с прозрачностью. Как GIF, так и PNG. А вот когда добавляю программно - не помогают никакие ухищрения.
---
не, не помогло, к сожалению. Спасибо в любом случае!
---
что не помогло? Какого формата картинки? Видимо 1с ну совсем не понимает картинки, рожденные онлайн редактором. Может попробовать что-то другое офлайновое, типа paint.net?
---
Картинки в PNG либо GIF.
Я подчёркиваю, что если добавляю ту же самую картинку в конфигураторе на ТД, то всё ОК с прозрачностью. Как GIF, так и PNG. А вот когда добавляю программно - не помогают никакие ухищрения.
---
У меня вот так сделано, и вроде все работает

```1c
    Рисунок=ТабличныйДокумент.Рисунки.Добавить(ТипРисункаТабличногоДокумента.Картинка);
    Рисунок.ЦветФона=Новый Цвет();
    Рисунок.Картинка = Новый Картинка(РезНастройка.Ссылка.Картинка.получить()); // картинка в хранилище в справочнике
    Рисунок.Лево = РезНастройка.Лево;//3
    Рисунок.Верх = РезНастройка.Верх;//ТабличныйДокумент.ВысотаСтраницы-80;
    Рисунок.Высота = РезНастройка.Высота;//50;
    Рисунок.Ширина = РезНастройка.Ширина;//90;
    Рисунок.РазмерКартинки =РазмерКартинки.Растянуть;
    Рисунок.Линия= Новый Линия(ТипЛинииРисункаТабличногоДокумента.НетЛинии);
```

Может надо именно Рисунок.ЦветФона=Новый Цвет(); указать
---
Браво, маэстро! Это решение:
Рисунок.ЦветФона=Новый Цвет();